

GL00309

FC
USGS
OFR
79-1641

UNITED STATES DEPARTMENT OF THE INTERIOR
GEOLOGICAL SURVEY

A Program in Hewlett-Packard BASIC for X-Y
Plotting and Line-Fitting of Isotopic and Other Data

Kenneth R. Ludwig
Denver, Colorado 80225

Open-file Report 79-1641

1979

**UNIVERSITY OF UTAH
RESEARCH INSTITUTE
EARTH SCIENCE LAB.**

This report is preliminary and has not been edited or reviewed for conformity with U. S. Geological Survey standards and nomenclature.

A Program in Hewlett-Packard BASIC for X-Y plotting
and line-fitting of isotopic and other data

by

Kenneth R. Ludwig

INTRODUCTION

The following program is intended mainly for use with isotope-ratio data, such as Rb-Sr isochron diagrams, but is general enough in concept to be useful for X-Y plotting and linear regression calculation of virtually any type of data. As given here, the programs were written for a Hewlett-Packard model 9831 desktop computer with 12K words of memory and a 9872 plotter,^{1/} and can be readily modified for use with Hewlett-Packard 9830/9862 computers/plotters. The resulting plots can be user scaled to different sizes and height/width ratios, and are high enough in drafting quality that no difficulty has been experienced in acceptance as camera-ready diagrams by scientific journals.

FEATURES

The plotting program will accept data from both a 50x35 data array (Ludwig, 1979) and from keyboard input. To have the plot boundaries, axis ticks, tick labels and axis labels drawn, the user need specify only the X and Y limits of the plot, the X and Y labels, and the dimensions of the plot. Tick intervals and adjustments for different height/width ratios of the plot are made automatically, as are centering and right-justification of the tick labels. Data points may be plotted as keyboard symbols (such as *, X, +, B); as open or solid regular polygons with user specified number of sides, (squares, triangles, circles, etc.), relative size, and rotational orientation; as

^{1/} Trade names are for descriptive purposes only, and do not constitute endorsement by the U.S. Geological Survey.

crosses or boxes (open or solid) whose X and Y dimensions correspond to analytical error; or as open or solid error ellipses. Once plotted, a least-squares regression line may be calculated using a modified York (1969) algorithm, which assumes error in both X and Y parameters, and permits weighting of the points according to their analytical errors and error correlations (for U-Pb isotope data, see Ludwig, in press). The standard (Model 1) York treatment assumes all scatter from a straight line is due to normally-distributed analytical error. If such treatment yields residuals indicating less than 30 percent probability of the observed scatter resulting from analytical error alone, the best-fit line is recalculated assuming either that all the scatter is due to some unknown mechanism yielding a normal distribution of residuals (Model 2), or assuming that the scatter is due to a combination of X and Y analytical error plus an "external" variation in Y values (Model 3). Model 2 assumes that nothing is known about the reason for the scatter, so all points are weighted equally without regard to analytical error. Model 3 is especially useful for isochron diagrams of the Rb-Sr type, in that it assumes a variable (and normally distributed) initial-Y value for the data points (similar to the Model 3 solution of McIntyre and others, 1966). The Model 3 solution of this report also solves for the most probable value of the standard deviation of the initial-Y value by forcing the sums of the squared residuals to equal $N-2$ (N = number of data points).

Once a Yorkfit line has been calculated, the user may have an isochron age and associated uncertainties calculated, for any isotope ratio including $^{207}\text{Pb}/^{204}\text{Pb}$ – $^{206}\text{Pb}/^{204}\text{Pb}$.

To simplify preparation of camera-ready plots, a lettering sub-routine is included that permits lettering of phrases anywhere on the plot with letters of arbitrary size, slant, and line rotation.

Special features of these programs are that,

- (a) The user may scale the plot to arbitrary dimensions (e.g. 10 in. high by 2 in. wide, 2 in. high by 14 in. wide) without distortion of letters or symbols.
- (b) Tick labels are automatically centered (along the X axis), right-justified (along the Y axis) and lettered with a constant number of digits to the right of the decimal point.
- (c) Keyboard data-input is extremely simple, with automatic acceptance of either X and Y values only, X and Y values with errors and error-correlations (required for error-ellipse plotting and Yorkfit input, or X and Y values with both errors and error-correlations).

Sample plots are shown in figures 1-3. Instructions for running the program are printed in appendix 1.

The program is structured into four segments occupying four files on the standard Hewlett-Packard data cartridge. Part 1 (appendix II) draws the basic plot box, with tick and axis labels; Part 2 (appendix III) acquires and plots the data, letters phrases, and calculates isochron ages; Part 3 (appendix IV) is the modified York (1969) algorithm for calculating best-fit lines; Part 4 (appendix V) contains instructions for the program. The file locations of these segments must be specified in lines 120-190 of Part 1.

The program can be copied onto blank data cartridges if the appropriate cartridge is supplied to the author, Box 25046, Denver, CO 80225. Simpler versions requiring keyboard input only and for Hewlett-Packard 9830/9862 computer/plotters are also available.

REFERENCES

- Ludwig, K.R., 1979, A generalized program in Hewlett-Packard BASIC for creating, editing, and printing out data files for geochemical applications; U. S. Geological Survey Open-File Report 79-1434, 24 p.
- Ludwig, K.R., in press, Calculation of uncertainties of U-Pb isotope data, Earth Planet. Sci. Lett.
- McIntyre, G.A., Brooks, C., Compston, W., and Turek, A., 1966, The statistical assessment of Rb-Sr isochrons, Jour. Geophys. Res., v. 71, p. 5459-5468.
- York, Derek, 1969, Least squares fitting of a straight line with correlated errors; Earth Planet. Sci. Lett., v. 5, p. 320-324.

FIGURE CAPTIONS

Fig. 1: Example of standard-size plot. From left to right, symbols are: solid, 6-sided polygon; solid error-ellipse; error-cross; asterisk; solid error-rectangle; open 15-sided polygon; open error-ellipse. Dashed line is the Model 3 Yorkfit solution.

Fig. 2: Example of plot with arbitrary dimensions. All symbols are solid error-ellipses, dashed line is Model 1 Yorkfit solution.

Fig. 3: Example of small-size plot, photographically reduced from 5x7 in. size to fit in a journal with 3 in. column-width. Symbols are open and solid 20-sided polygons, solid 3-sided polygons rotated to 270°, and crosses ("plus" symbol on keyboard).

$^{207}\text{Pb}/^{204}\text{Pb}$




Figure 1.




FIGURE 2




FIGURE 3

INSTRUCTIONS FOR K.R. LUDWIG HP-9831/9872 GENERAL PLOTTING PROGRAM
U.S. Geological Survey Open-File Rept. #79-1641INTRODUCTION

This program is written for an HP-9831 computer with 12K words memory, and an HP-9872 plotter. Data may be entered either from the keyboard, or from data stored in a split-precision, 50X35 array as created by the K.R.L. data-file program (U.S.G.S. Open-File Rept. # 79-1434). X- and Y-axis labels may either be taken from the parameter names defined in the data-file, or keyboard-input. Data-points may be plotted using a variety of symbols, including open/solid polygons, error-boxes, error-crosses, or error-ellipses. Errors and error-correlations may be input either from a data-file, or from the keyboard, even if the data are from a data-file. Plot-size and relative dimensions may be adjusted by the user over a wide range without distortion of symbols. Once plotted, a least-squares regression line may be calculated and plotted, using the standard or one of two modified York-type algorithms. Isochron ages and errors for most isotope-ratio plots (including 207-Pb/204-Pb vs. 206-Pb/204-Pb) may be calculated from the regression-line data.

DISPLAY
-----RESPONSE

DATA FROM TRACK, FILE#?

If you wish to plot data from a data-file, enter the track and file numbers of that data-file. If data are to be entered only from the keyboard, enter a space.

SIZE: 1=STD, (2)=SMALL, 3=INSTR.?

Enter 1 for a plot scaled to fit on a horizontal, 8 x 10.5 in. sheet of paper; enter 2 for a smaller plot expressly scaled for reproduction as a 35mm transparency or journal-figure (or -2 for a plot with arbitrary size but relatively larger labels and symbols); enter 3 to obtain a printout of these instructions. Any other number permits the user to adjust the size of the plot, using the P1-P2 controls of the plotter.

X-AXIS (NAME OR PARAMETER #)? If data are to be taken mainly from a data-file, enter the parameter number of the X-axis data. Otherwise, enter a label from the keyboard. If you wish all numerals of the label to be superscripts, enter ^ as the first character of the label (e.g., ^87Rb/86Sr). If you wish the numerals to be subscripts, enter an asterisk (*) as the first character (e.g., *Al2O3).

Y-AXIS (NAME OR PARAMETER #)? Same input as for X-axis.

X AND Y LIMITS? Enter 4 values: X-MIN, X-MAX, Y-MIN, Y-MAX, to define the limits of the plot.

A list summarizing the symbol options now appears as printout.

An input of '@' will restart the plot set-up part of the program.

An input of 'p' or 'P' will permit any regular polygon of any orientation to be used as a plotting-symbol. The later 'ROTATION' request asks for the angle in degrees (3 o'clock=0 degrees, 12 o'clock=90 degrees) of a vertex of the polygon, 'SIZE' request asks for relative size of the polygon. Values of 2 to 3 are most useful. Size is proportional to the number entered. P gives solid, p open symbols.

An input of 'e' or 'E' will give open or solid error-rectangles as plotting-symbols.

An input of 'c' or 'C' will give open or solid error-ellipses as plotting-symbols. You will need to know the correlation-coefficient of the X and Y errors (0 if uncorrelated).

An input of '.' will give error-crosses as plotting symbols.

An input of 'L' or 'l' will permit you to letter any statement anywhere on the plot. The subsequent 'SIZE (.5-2), (ANGLE), (SLANT)?' display requests a number defining the relative size of the letters (0.5 to 2 is the most useful range); the angle of the line along which the phrase is to be lettered, in degrees; and the slant of the letters (also in degrees). Note that input of the latter two parameters is optional -- only the letter-size parameter, or the

letter-size plus line-angle parameters need be entered. The next display will be, 'POSITION PEN. (LIVE KEYBOARD)'. At this point, you may position the plotter-pen by using the four 'arrow' keys at the top of the keyboard. DO NOT PRESS A CHARACTER KEY, or the character will be immediately printed on the plot. Press the STOP key when the pen is positioned correctly. Then type in the statement to be lettered, and press the 'EXECUTE' key. To have the phrase lettered on successively lower lines, include the two-character symbol '// wherever you wish the lettering of the phrase to be repositioned one line lower than before, starting just below the first character of the phrase. If the three-character symbol '+/-' is included in the phrase, it will appear as a '+' over a '-' symbol, as is usually drafted, and will occupy one character space.

An input of 'A' or 'a' will result in the calculation of an isochron age from a just-calculated Yorkfit slope. Enter the appropriate decay constant (in units of reciprocal years) when requested. However, if the plot is 207-Pb/204-Pb vs 206-Pb/204-Pb, a 207/206 age will be calculated using the IUGS decay constants (207 and 204 must appear in the Y-label and 206 and 204 in the X-label).

An input of '^' will result in a line connecting the data-points, in the order that they were entered.

Any other UPPER CASE keyboard symbol may also be used as a plotting-symbol (such as X, *, +, O). Lower-case symbols and other asymmetrical characters will not be plotted accurately, however.

DISPLAY

RESPONSE

X, (ERR,) Y (,ERR) (,CORRELATION)? If keyboard input, the data for each point may be entered as just the X and Y values, (2 values separated by a comma), as the X and Y values together with their errors (X, X-error, Y,Y-error) or as the X and Y values together with their errors and with error-correlations (X, X-error, Y, Y-error, error-correl.). If error or error-correlation information that was not entered is needed for later plotting or Yorkfit purposes, it will be requested later.

If the data is to be input from a data-file, input a asterisk followed by the number(s) of the sets to be plotted.

An input of *8 will input set 8, whereas an input of *9,42 would input sets 9 through 42, inclusive, from the data-file in memory. If you wish to change the data-file in memory, input 'DF' or 'df' and the display will request the file number of the new data-file.

When all data for the particular plotting-symbol has been entered, enter a space.

?EDIT ENTRY#? (0 IF OK)?

If your data was input correctly, enter 0. If corrections are necessary, enter the number of the set to be edited.

DATA TO BE RETAINED (1=YES)?

If you wish the just-plotted data to be combined with subsequently input data (perhaps with different plotting symbols) for a later Yorkfit, input 1. Otherwise, input any other number.

X-ERRORS, Y-ERRORS (LAST BLOCK)

This display appears if required error-information has not been entered. For keyboard error-input, enter the X- and Y-errors, separated by a comma. If the errors are stored in the data-file, however, input an asterisk, followed by the parameter numbers of the X- and Y-errors (e.g., *10,16).

NOTE: ALL ERRORS MUST BE INPUT IN PERCENT AND, FOR YORKFIT INPUT, AT THE 2-SIGMA LEVEL.

ERROR-CORRELATION (LAST BLOCK)?

Similar response to above.

NAME FOR THIS DATA SET?

Enter a name for the Yorkfit printout

REMARKS ON YORKFIT SOLUTIONS

The Yorkfit routine may be done in three ways. The first solution (York's original algorithm) weights the points according to their analytical errors, and calculates the best-fit line and errors assuming that the only cause of scatter from a straight line is analytical error. However, this may not be realistic if the calculated probability (from the comparison of the sums of the squares of the residuals (SUMS) with the degrees of freedom) is low. This probability corresponds to how often one may expect the data, with their associated analytical errors, to give the observed amount of scatter. If this probability is less than 30%, the Yorkfit will be redone by assuming either equally-weighted points with zero error-correlations (model 2), or by assuming that the cause of the excess scatter is due to an additional variance in the Y-parameter (model 3). The model 2 solution assumes only that the cause for scatter has a normal distribution. Model 3 corresponds to the variable initial-ratio problem of Rb-Sr isochrons. The model 3 solution will be calculated whenever the Y-axis name includes the characters 87Sr/86Sr, 143Nd/144Nd, 206Pb/204Pb, or 207Pb/204Pb when the X-axis is not 206Pb/204Pb. The solution will also calculate the most probable value of the standard-deviation of the initial Y. If the model 1 probability is greater than 30%, only the model 1 solution will be solved. If the probability is between 1% and 30%, the user can choose which solution to use. If the probability is less than 1%, only the model 2 or 3 solution can be used (for plotting of the best-fit line and for later isochron-age solutions). If a model 2 solution is obtained when a model 3 solution is desired (or vice-versa), the correct model-solution can be obtained by executing 'SFLAG1' before running the program.

APPENDIX II -- LIST OF PART 1 OF PROGRAM

```

10 COM AI[50,18],CI[35,13],DS[50,35],I2,M,B$[80]
20 REM PART 1 (SET-UP PLOT) OF K.R LUDWIG GENERAL PLOTTER PROGRAM
30 REM U.S.GEOLOGICAL SURVEY OPEN-FILE REPT. #79-1641
40 DIM A$[38],E$[25],F$[25],V[50],XS[50],YS[50],PS[50],I$[14],J$[14]
50 DIM QS[50],US[50],MS[50],ZS[50],RS[50],T$[45],Z$[12],H$[25],TS[50],SS[50]
60 DIM ES[50],NS[50],FS[50],G$[1],II[4,2]
70 Z$="0123456789"
80 DEG
90 P1=H1=L1=L2=0
100 REM TRACK #S - I(n,2) - AND FILE #S - I(n,1) - OF PROGRAM SEGMENTS
110 REM FOR I(k,n), k IS THE SEGMENT #
120 I[1,1]=0
130 I[1,2]=1
140 I[2,1]=0
150 I[2,2]=2
160 I[3,1]=0
170 I[3,2]=3
180 I[4,1]=0
190 I[4,2]=4
200 SLANT 0
210 GOTO 580
220 FIXED 0
230 RETURN
240 FIXED 1
250 RETURN
260 FIXED 2
270 RETURN
280 FIXED 3
290 RETURN
300 FIXED 4
310 RETURN
320 FIXED 5
330 RETURN
340 DEF FNA(X)=INTLG TABS X
350 DEF FNB(X)
360 Q=10^FNAX/8
370 IF ABS(X/Q) <= 12 THEN 400
380 Q=2*Q
390 GOTO 370
400 A=ABSQ/10^FNAQ
410 IF A=INTA THEN 430
420 Q=INTA*10^FNAQ
430 RETURN Q
440 DEF FNF(X)
450 S=E=0
460 N=X
470 N=10*(ABSN-INTABSN)
480 IF N=0 THEN 520
490 S=S+1
500 E=1

```

```

510 GOTO 470
520 RETURN S
530 DEF FNK(X)=1+(X#INTX)+(ABSX >= 1)*FNA(ABSX+ NOT X)+FNFX
540 DEF FNU(J)=J+POS(T$,"")
550 DEF FNC(I)=POS(Z$,A$[I,I])
560 DEF FNQ(I)=FNC1*(LEN(A$)=1) OR FNC1*FNC2*(LEN(A$)=2)
570 DEF FNJ(X)=- ( NOT X)-(X#0)*(FNKX+J-FNFX+(X=INTX)*(J#0))
580 B$=" "
590 DISP "DATA FROM TRACK#, FILE#";
600 INPUT T$
610 IF T$=" " THEN 640
620 IF NOT POS(Z$[1,2],T$[1,1]) OR NOT POS(Z$,T$[FNU1,FNU1]) THEN 590
630 LOAD DATA #VAL(T$),VAL(T$[FNU1])
640 IF B$=" " THEN 670
650 PRINT LIN1,B$,LIN1,I2"SETS,"M"PARAMETERS DEFINED"LIN2
660 FIND #I[2,1],I[2,2]
670 DISP "SIZE: 1=STD, (2)=SMALL, 3=INSTR.";
680 INPUT F
690 IF F#3 THEN 710
700 LINK #I[4,1],I[4,2]
710 B=(ABSF=2)
720 IF F#1 AND F#2 THEN 740
730 MAP B*1500,1E+04-B*1500,1125*B,7500-B*1125
740 GOSUB 760
750 GOTO 1190
760 DISP "X-AXIS (NAME OR PARAMETER #";
770 INPUT A$
780 E$=A$
790 IF NOT FNQ0 THEN 870
800 L1=VAL(E$)
810 IF L1 >= 1 AND L1 <= M THEN 840
820 GOSUB 2260
830 GOTO 760
840 TRANSFER C[L1,1] TO A$
850 GOSUB 1000
860 E$=A$
870 DISP "Y-AXIS (NAME OR PARAMETER #)";
880 INPUT A$
890 F$=A$
900 IF NOT FNQ0 THEN 980
910 L2=VAL(F$)
920 IF L2 >= 1 AND L2 <= M THEN 950
930 GOSUB 2260
940 GOTO 870
950 TRANSFER C[L2,1] TO A$
960 GOSUB 1000
970 F$=A$
980 RETURN
990 REM SPACE-TRIM ROUTINE FOR DATA-FILE PARAMETER NAMES
1000 I$=A$[14,25]
1010 H$=A$[1,13]
1020 J$=I$
1030 GOSUB 1110
1040 I$=J$

```



```

1050 J$=H$
1060 GOSUB 1110
1070 A$=J$
1080 A$[1+LEN(A$)]=" "
1090 A$[1+LEN(A$)]=I$
1100 RETURN
1110 IF J$[1,1]#" " THEN 1150
1120 IF LEN(J$)=0 THEN 1180
1130 J$=J$[1+POS(J$," ")]
1140 GOTO 1110
1150 IF J$[LEN(J$)]#" " THEN 1180
1160 J$=J$[1,LEN(J$)-1]
1170 GOTO 1150
1180 RETURN
1190 DISP "X AND Y LIMITS";
1200 FIND #I[2,1],I[2,2]
1210 INPUT A1,B1,A2,B2
1220 REM CALCULATE TIC INTERVALS, MAX. # DIGITS IN TIC LABELS
1230 C5=(B1-A1)
1240 D5=(B2-A2)
1250 L=B1+C5/25
1260 A8=FNBC5
1270 A5=FNBD5
1280 FOR Y=A2 TO B2 STEP 2*A5
1290 IF ((Y<0)+FNKY)<P1 THEN 1310
1300 P1=(Y<0)+FNKY
1310 NEXT Y
1320 FOR X=A1 TO B1 STEP 2*A8
1330 IF ((X<0)+FNKY)<H1 THEN 1350
1340 H1=(X<0)+FNKX
1350 NEXT X
1360 REM CALCULATE PLOT DIMENSIONS & ROOM REQUIRED FOR LABELS
1370 WRITE (705,*)"OP"
1380 ENTER (705,*)X1,Y1,S2,H2
1390 S2=S2-X1
1400 H2=H2-Y1
1410 P=H2/S2
1420 P9=P*(P <= 1)*(0.85*(H2<7500)+0.65*(H2 >= 7500)*S2/7500)
1430 P9=P9+(P>1)*(0.85*(S2<7500)+0.65*(S2 >= 7500)*H2/7500)
1440 F3=1.9/P9+1.3*(B+(SQR(S2^2+H2^2)<6000))
1450 F4=1.5/P9+(B+(SQR(S2^2+H2^2)<6000))/2
1460 F=P*(F3*(0.6+(POS(F$[1,1],"^") OR POS(F$[1,1],"*"))/3)+P1*F4)/54
1470 P3=(A1-L*F)/(1-F)
1480 F=(F4+F3*(0.8+(POS(E$[1,1],"^")+POS(E$[1,1],"*"))/3))/43
1490 P7=(A2-B2*F)/(1-F)
1500 SCALE P3,L,P7,B2+D5/13
1510 REM DRAW PLOT-BOX
1520 WRITE (705,*)"TL.5"
1530 XAXIS A2,A8,A1,B1
1540 WRITE (705,*)"TL0,.5"
1550 YAXIS B1,A5,A2,B2
1560 WRITE (705,*)"TL.5"
1570 YAXIS A1,A5,A2,B2
1580 WRITE (705,*)"TL0,.5"

```

```

1590 XAXIS B2,A8,A1,B1
1600 IF B=0 THEN 1670
1610 PLOT B1+C5/400,B2+D5/300
1620 IPLOT -1.005*C5,0
1630 IPLOT 0,-(1+2/300)*D5
1640 IPLCT 1.005*C5,0
1650 IPLOT 0,(1+2/300)*D5,-1
1660 REM LABEL TICS
1670 J=0
1680 FOR X=A1 TO B1 STEP 2*(1+((H1>5) OR S2 <= 5000))*A8
1690 IF FNFX<J THEN 1710
1700 J=FNFX
1710 NEXT X
1720 FOR X=A1 TO B1 STEP 2*(1+((H1>5) OR S2 <= 5000))*A8
1730 LABEL (*,F4,1.7,0,P)
1740 PLOT X,A2,1
1750 CPLOT FNJX/2-1,-1
1760 GOSUB 1+J*(1- NOT X) OF 220,240,260,280,300,320
1770 LABEL (*)X;
1780 NEXT X
1790 J=0
1800 FOR Y=A2 TO B2 STEP 2*A5
1810 IF FNFY<J THEN 1830
1820 J=FNFY
1830 NEXT Y
1840 FOR Y=A2 TO B2 STEP 2*A5
1850 PLOT A1,Y,1
1860 CPLOT FNJY-1.3,-0.3
1870 GOSUB 1+J*(1- NOT Y) OF 220,240,260,280,300,320
1880 LABEL (*)Y;
1890 NEXT Y
1900 REM LABEL AXES
1910 PLOT A1+C5/2,A2,1
1920 CPLOT 0,-1
1930 LABEL (*,F3,1.7,0,P)
1940 CPLOT -(LEN(E$)-1)/2,-0.8-POS(E$[1,1],"^")/3
1950 H$=E$
1960 GOSUB 2050
1970 PLOT A1,A2+D5/2,1
1980 LABEL (*,F4,1.7,0,P)
1990 CPLOT -P1,0
2000 LABEL (*,F3,1.7,90,P)
2010 CPLOT -LEN(F$)/2,0.6+POS(F$[1,1],"*")/3
2020 H$=F$
2030 GOSUB 2050
2040 GOTO 2150
2050 J=POS(H$[1,1],"^")-POS(H$[1,1],"*")
2060 FOR I=1+ABSJ TO LEN(H$)
2070 IF NOT POS(Z$,H$[I,I]) THEN 2120
2080 CPLOT 0,J/3
2090 LABEL (*)H$[I,I];
2100 CPLOT 0,-J/3
2110 GOTO 2130
2120 LABEL (*)H$[I,I];

```

```
2130 NEXT I
2140 RETURN
2150 FIND #I[2,1],I[2,2]
2160 PRINT "DATA SYMBOL/CODE OPTIONS"LIN2
2165 PRINT "@ ----- TO REDEFINE/REDRAW PLOT-BOX"
2170 PRINT "E/e ---- SOLID/OPEN ERROR-RECTANGLE"
2180 PRINT "C/c ---- SOLID/OPEN ERROR-ELLIPSE"
2190 PRINT ". ----- ERROR-CROSS"
2200 PRINT "P/p ---- SOLID/OPEN POLYGON"
2210 PRINT "^ ----- LINE CONNECTING PLOTTED POINTS"
2220 PRINT "L/l ---- TO LETTEP ON PLOT"
2230 PRINT "A/a ---- TO CALCULATE AN ISOCHRON AGE"
2240 PRINT "+ * X . ANY OTHER KEYBOARD SYMBOL WILL PLOT AS INPUT"LIN1
2250 LINK #I[2,1],I[2,2]
2260 PRINT LIN2"ONLY" "M" PARAMETERS DEFINED IN THIS DATA-FILE"LIN2
2270 RETURN
2280 END
```

APPENDIX III -- LIST OF PART 2 OF PROGRAM

```

10 REM PART 2, GENERAL PLOTTER, U.S. GEOLOGICAL SURVEY OPEN-FILE REPT. # 79- 1641
20 GOTO 130
30 DEF FNU(J)=J+POS(T$,",")
40 DEF FNP(J)
50 IF FNU0 THEN 70
60 T$[1+LEN(T$)]=","0"
70 A$=T$
80 T$=T$[FNU1]
90 IF POS(Z$,A$[J,J]) THEN 110
100 RETURN 0
110 RETURN VAL(A$[J])
120 DEF FNO(F)=Y[I]+(-B3*X+F*SQPF1)/(2*S3)
125 LOAD #I[1,1],I[1,2]
130 Z$="-.0123456789"
140 PLOT B1,B2,1
150 K=1
160 N=0
170 STANDARD
180 A$="@LlAa^PpEeCc.+*"
190 DISP "DATA SYMBOL OR CODE";
200 INPUT G$[1,1]
210 G=POS(A$,G$[1,1])
220 GOTO G OF 125,2380,2380,2650,2650,1830,1290,1290
230 PRINT "USE * PREFIX FOR DATA-FILE SET INPUT"LIN1
240 PRINT "ENTRY#"TAB8"SET#"TAB16;E$[1,10];TAB28"% ERR"TAB44;F$[1,10];TAB56;
250 PRINT "% ERR"TAB65;"CORR"LIN2
260 REM DATA-INPUT
270 GOSUB 290
280 GOTO 330
290 DISP "X,(ERR,) Y,(ERR,)(CORRELATION)";
300 FIND #I[3,1],I[3,2]
310 INPUT T$
320 RETURN
330 IF T$=" " THEN 1210
340 IF T$[1,1]="*" THEN 580
350 IF POS(T$,"DF") OR POS(T$,"df") THEN 500
360 REM KEYBOARD DATA-INPUT ROUTINE
370 I=N=1+N
380 GOSUB 400
390 GOTO 270
400 X[I]=FNP1
410 IF FNU0#0 THEN 440
420 U[I]=0
430 GOTO 450
440 U[I]=FNP1
450 Y[I]=FNP1
460 S[I]=FNP1
470 R[I]=FNP1
480 PRINT I;TAB16;X[I];TAB28;U[I];TAB44;Y[I];TAB56;S[I];TAB65;R[I]
490 RETURN

```

```

500 T$=T$[3]
510 DISP "DATA FROM TRACK#, FILE #";
520 INPUT T,F
530 LCAD DATA #T,F
540 PRINT LIN1,B$,LIN1,I2"SETS,"M"PARAMETERS DEFINED"LIN2
550 GOTO 270
560 GOTO 230
570 REM DATA-FILE DATA-INPUT ROUTINE
580 T$=T$[1+POS(T$,"*")]
590 IF T$="0" THEN 1210
600 S1=FNPI
610 S2=FNPI
620 S2=S2+S1* NOT S2
630 S1=S1*(S1 >= 1)*(S1 <= I2)+(S1<1)+I2*(S1>I2)
640 S2=S2*(S2 >= 1)*(S2 <= I2)+(S2<1)+I2*(S2>I2)
650 N=1+N+S2-S1
660 FOR I=N-(S2-S1) TO N
670 Z[I]=I-N+S2
680 GOSUE 700
690 GOTO 750
700 X[I]=D[Z[I],L1]
710 Y[I]=D[Z[I],L2]
720 U[I]=S[I]=R[I]=0
730 PRINT I;TAB8;Z[I];TAB16;X[I];TAB44;Y[I]
740 RETURN
750 NEXT I
760 PRINT LIN1
770 GOTO 270
780 GOSUE 2050
790 IF A=0 THEN 810
800 GOSUE 1010
810 IF G#10 AND G#11 OF J=0 THEN 830
820 GOSUE 2120
830 FOR I=K TO N
840 IF (X[I] >= A1)*(X[I] <= B1)*(Y[I] >= A2)*(Y[I] <= B2) THEN 870
850 PRINT "SET" I "LIES OUTSIDE PLOT LIMITS" LIN1
860 GOTO 970
870 PLOT X[I],Y[I],1
880 GOTO G-6 OF 1320,1320,1590,1590,1590,1590,1590,1590,940,940
890 REM KEYBOARD-SYMBOL DATA PLOTTING
900 LABEL (*,1.1+B/2,1,0,P)
910 CPLOT -1/3,-1/4
920 LABEL (*)G$[1,1];
930 GOTO 970
940 LABEL (*,3*(1+B)/2,1.7,0,P)
950 GOTO 910
960 PEN
970 NEXT I
980 PLOT B1,B2,1
990 GOTO 1870
1000 REM ERROR-INPUT
1010 DISP "X-ERRORS, Y-ERRORS (LAST BLOCK)";
1020 INPUT I$
1030 IF T$[1,1]="*" THEN 1110

```

```

1040 F=FNPI
1050 F4=FNPI
1060 FOR I=K TO N
1070 U[I]=F
1080 S[I]=F4
1090 NEXT I
1100 RETURN
1110 P1=FNPI
1120 P2=FNPI
1130 IF P1 >= 1 AND P1 <= M AND P2 >= 1 AND P2 <= M THEN 1160
1140 GOSUB 2960
1150 GOTO 1010
1160 FOR I=K TO N
1170 U[I]=D[Z[I],P1]
1180 S[I]=D[Z[I],P2]
1190 NEXT I
1200 RETURN
1210 PRINT LIN1
1220 DISP "?EDIT ENTRY#? (0 IF OK)";
1230 INPUT I
1240 GOTO NOT I*(1+(G>8)*(G<14)) OF 830,780
1250 GOSUB 290
1260 GOSUB 400
1270 GOTO 1210
1280 REM POLYGON-SYMBOL PLOT ROUTINE
1290 DISP "# SIDES, ROTATION, SIZE (1-10)";
1300 INPUT B3,A9,S3
1310 GOTO 230
1320 FOR U=0 TO (S3>2.5)+50*(G=7)
1330 A7=(L-P3)*(S3/340-U/900)
1340 IF A7<0 THEN 960
1350 FOR J=A9 TO A9+360 STEP 360/B3
1360 PLOT X[I]+A7*COSJ,Y[I]+A7*SINJ*(B2+D5/13-P7)/(P*(L-P3))
1370 NEXT J
1380 NEXT U
1390 GOTO 960
1400 REM ERROR-ELLIPSE PLOT ROUTINE
1410 U=1-R[I]^2
1420 B3=-2*R[I]/(P[I]*Q[I]*U)
1430 S3=1/(U*Q[I]^2)
1440 FOR J=1 TO 1+(G=11)
1450 FOR F=-1 TO 1 STEP 2
1460 F4=-F*P[I]/(1+INT(600*P[I]/(L-P3)))
1470 FOR X=F*P[I] TO -F*P[I]+F4/2 STEP F4
1480 F1=(B3*X)^2-4*S3*(X*X/(U*P[I]^2)-1)
1490 F1=F1*(F1>0)
1500 PLOT X[I]+X,FNOF
1510 IF J=1 THEN 1540
1520 F=-F
1530 PLOT X[I]+X,FNOF
1540 NEXT X
1550 IF J=2 THEN 960
1560 NEXT F
1570 NEXT J

```

```

1580 GOTO 960
1590 IF X[I]*U[I]*Y[I]*S[I] THEN 1620
1600 PRINT LIN2"VALUES AND ERRORS MUST BE NONZERO FOR THIS PLOT SYMBOL"LIN3
1610 GOTO 190
1620 P[I]=X[I]*U[I]/100
1630 Q[I]=Y[I]*S[I]/100
1640 GOTO G-8 OF 1660,1660,1410,1410,1770
1650 REM ERROR-BOX PLOT ROUTINE
1660 IPLOT -P[I],Q[I]
1670 IPLOT 0,-2*Q[I]
1680 IPLOT 2*P[I],0
1690 IPLOT 0,2*Q[I]
1700 IPLOT -2*P[I],0
1710 IF G=10 THEN 960
1720 FOR J=1 TO 3210*P[I]/(L-P3)
1730 IPLOT (J/2=INT(J/2))* (L-P3)/800,Q[I]*2*SIN(90*(J-2))
1740 NEXT J
1750 GOTO 960
1760 REM ERROR-CROSS PLOT ROUTINE
1770 IPLOT 0,Q[I]
1780 IPLOT 0,-2*Q[I],-1
1790 IPLOT -P[I],Q[I]
1800 IPLOT 2*P[I],0,-1
1810 GOTO 970
1820 REM PLOT LINE THROUGH DATA-POINTS ROUTINE
1830 FOR I=1 TO N
1840 PLOT X[I],Y[I]
1850 NEXT I
1860 GOTO 140
1870 IF N<2 THEN 1930
1880 DISP "YORKFIT? 1=YES";
1890 INPUT F
1900 IF F#1 THEN 1930
1910 GOSUB 1990
1920 GOTO 2360
1930 DISP "DATA TO BE RETAINED? (1=YES) ";
1940 INPUT J
1950 IF J#1 THEN 150
1960 GOSUB 1990
1970 K=1+N
1980 GOTO 180
1990 GOSUB 2050
2000 IF A=0 THEN 2020
2010 GOSUB 1010
2020 IF J=0 OR G=11 OR G=12 THEN 2100
2030 GOSUB 2120
2040 RETURN
2050 A=J=0
2060 FOR I=K TO N
2070 A=A+ NOT U[I]+ NOT S[I]
2080 J=J+ NOT R[I]
2090 NEXT I
2100 RETURN
2110 REM ERROR-CORRELATION INPUT ROUTINE

```

```

2120 DISP "ERROR CORRELATION (LAST BLOCK)";
2130 INPUT T$
2140 IF T$[1,1]#"*" THEN 2280
2150 F4=FN P2
2160 IF F4 >= 1 AND F4 <= M THEN 2190
2170 GOSUB 2960
2180 GOTO 2120
2190 FOR I=K TO N
2200 F=D[Z[I],F4]
2210 IF F >= -1 AND F <= 1 THEN 2250
2220 PRINT LIN2"CHECK PARAMETER # -- ";
2230 PRINT "ERR. CORR. MUST BE BETWEEN -1 & 1."LIN3
2240 GOTO 2120
2250 R[I]=F
2260 NEXT I
2270 RETURN
2280 F=FN P1
2290 IF F<-1 OR F>1 THEN 2230
2300 FOR I=K TO N
2310 R[I]=F
2320 NEXT I
2330 RETURN
2340 K=1+N
2350 IF F#1 THEN 170
2360 LINK #I[3,1],I[3,2]
2370 REM LETTERING ROUTINE
2380 DISP "SIZE (.5-2), (ANGLE), (SLANT)";
2390 INPUT T$
2400 F=FN P1
2410 S1=FN P1
2420 S2=FN P1
2430 SLANT S2
2440 LABEL (*,F*F3,1.7,S1,P)
2450 DISP "POSITION PEN. (LIVE KEYBOARD"
2460 LETTER
2470 DISP "INPUT STATEMENT FOR LETTERING";
2480 INPUT T$
2490 Q0=0
2500 FOR I=1 TO LEN(T$)
2510 IF T$[I,I+2]#"+" THEN 2550
2520 WRITE (705,2800)
2530 Q0=1+Q0
2540 I=3+I
2550 IF T$[I,1+I]#"//" THEN 2590
2560 CPLOT -Q0,-1
2570 Q0=0
2580 I=2+I
2590 LABEL (*)T$[I,I];
2600 Q0=1+Q0
2610 NEXT I
2620 SLANT 0
2630 GOTO 150
2640 REM ISOCHRON-AGE CALCULATION
2650 IF POS(E$,"206")*POS(E$,"204")*POS(F$,"207")*POS(F$,"204") THEN 2750

```



```

2660 DISP "DECAY CONSTANT - (1/yr)";
2670 INPUT A
2680 D=1E-06*LOG(1+Q)/A
2690 S3=1E-06*P2/(A*(1+Q))
2700 PRINT "MODEL"10;"ISOCHRON AGE =";
2710 FORMAT F8.2," +/-",F8.2," M.Y. (DECAY CONST. =",E11.4,"/YR)."
2720 WRITE (2,2710)D,S3,A
2730 WRITE (2,2790)
2740 GOTO 140
2750 H1=1
2760 D=FNZQ/1E+06
2770 S3=H1*(FNZ(Q+P2)-FNZ(Q-P2))/2E+06
2780 GOTO 2700
2790 FORMAT /,90"-",/,/,
2800 FORMAT "UC-99,0,0,99,4,0,-99,-2,9,99,0,-6,-99,-2,3,99,4,0,-99"
2810 REM FUNCTION FOR 207/206 AGE CALCULATION
2820 DEF FNZ(Q)
2830 A=9.8485E-10
2840 C=1.55125E-10
2850 IF Q>0.0156 AND Q<6.6 THEN 2890
2860 PRINT LIN2"CANNOT CALCULATE A 7/6 AGE FOR 7/6="Q,LIN2
2870 H1=0
2880 RETURN 0
2890 S=3E+09*(SGN(-1/2+(Q>A/(C*137.88)))+(Q>0.7))
2900 F=(C/A)*EXP(S*(C-A))
2910 E=LOG(1+(EXP(S*C)-1-F*(EXP(S*A)-1))/(1/(137.88*Q)-F))/A
2920 IF ABS(E-S)<1000 THEN 2950
2930 S=E
2940 GOTO 2900
2950 RETURN E
2960 PRINT LIN2"**** PARAMETER NUMBERS MUST BE BEWEEN 1 AND"M"****"LIN3
2970 RETURN
2980 END

```



```

510 FOR I=1 TO N
520 GOTO NOT I3+2* NOT Y5 OF 610,570,610
530 N[I]=(Y[I]*S[I]/200)^2
540 F[I]=R[I]*SQR(N[I]/(N[I]+S8^2))
550 Q[I]=1/(N[I]+S8^2)
560 GOTO 710
570 P[I]=1
580 Q[I]=1/Q^2
590 R[I]=0
600 GOTO 710
610 P[I]=(200/(X[I]*U[I]))^2
620 Q[I]=(200/(Y[I]*S[I]))^2
630 F[I]=R[I]
640 PRINT I;TAB9;X[I];TAB23;
650 FIXED 3
660 PRINT U[I];TAB33;
670 STANDARD
680 PRINT Y[I];TAB47;
690 FIXED 3
700 PRINT S[I];TAB63;R[I]
710 STANDARD
720 NEXT I
730 S1=B5=D=S2=E=Z4=X1=Y1=0.
740 FOR I=1 TO N
750 M[I]=SQR(P[I]*Q[I])
760 Z[I]=P[I]*Q[I]/(Q*Q*Q[I]+P[I]-2*Q*F[I]*M[I])
770 S1=S1+Z[I]
780 NEXT I
790 FOR I=1 TO N
800 X1=X1+Z[I]*X[I]/S1
810 Y1=Y1+Z[I]*Y[I]/S1
820 NEXT I
830 FOR I=1 TO N
840 U[I]=X[I]-X1
850 V[I]=Y[I]-Y1
860 B5=B5+(Z[I]^2)*((U[I]^2)/Q[I]-(V[I]^2)/P[I])
870 D=D+(Z[I]^2)*(U[I]*V[I]/P[I]-F[I]*(U[I]^2)/M[I])
880 Z4=Z4+Z[I]*U[I]*U[I]
890 S2=S2+Z[I]*X[I]*X[I]
900 E=E+(Z[I]^2)*(U[I]*V[I]/Q[I]-F[I]*(V[I]^2)/M[I])
910 NEXT I
920 S3=B5^2+4*D*E
930 IF S3<0 THEN 2310
940 S3=(-B5+SQR(S3))/(2*D)
950 FIND #I[2,1],I[2,2]
960 IF (ABS(S3-Q)>ABS(S3/1E+04)) THEN 430
970 Q=S3
980 B5=A8=A9=Z4=H1=H2=S2=A=C=F=F1=Z2=Z3=A5=D=E=S=F5=0
990 I9=1+I9
1000 IF I9>30 THEN 2310
1010 FOR I=1 TO N
1020 S=S+Z[I]*(V[I]-Q*U[I])^2
1030 NEXT I
1040 A7=Y1-Q*X1

```

```

1050 IF NOT Y5 OR NOT I3 OR Y5*(ABS(S-N+2)<0.01) THEN 1080
1060 S8=S8*SQR(S/(N-2))
1070 GOTO 410
1080 DISP "CALCULATING ERRORS"
1090 FOR I=1 TO N
1100 N[I]=Z[I]^2*(Q^2*V[I]/P[I]-2*Q^2*R[I]*U[I]/M[I]+2*Q*U[I]/Q[I]-V[I]/Q[I])
1110 E[I]=Z[I]^2*(Q^2*U[I]/P[I]-2*Q*V[I]/P[I]-U[I]/Q[I]+2*R[I]*V[I]/M[I])
1120 Z2=Z2+N[I]
1130 Z3=Z3+E[I]
1140 A=Z[I]^3*(F[I]*M[I]-Q*Q[I])*(Q*U[I]-V[I])
1150 C=U[I]/Q[I]+Q*V[I]/P[I]-F[I]*(Q*U[I]+V[I])/M[I]
1160 A5=A5+4*(A*C)/M[I]^2
1170 D4=2*Z[I]^2*(Q*Q[I]-F[I]*M[I])/(S1*M[I]^2)
1180 D=D+D4*U[I]
1190 E=E+D4*V[I]
1200 B5=B5+Z[I]^2*(U[I]^2*(1/Q[I]-2*Q*F[I]/M[I])+V[I]*(2*Q*U[I]-V[I])/P[I])
1210 NEXT I
1220 FOR I=1 TO N
1230 T[I]=N[I]-Z[I]*Z2/S1
1240 A8=A8+T[I]^2/P[I]
1250 E[I]=E[I]-Z[I]*Z3/S1
1260 A=D*(Q*Q*V[I]/P[I]+2*Q*U[I]/Q[I]-V[I]/Q[I])
1270 A=A+E*(Q*Q*U[I]/P[I]-2*Q*V[I]/P[I]-U[I]/Q[I])
1280 A=A-2*F[I]*(Q*Q*U[I]*D-V[I]*E)/M[I]
1290 F1=F1+Z[I]^2*A
1300 A9=A9+E[I]^2/Q[I]
1310 NEXT I
1320 D9=B5+A5+F1
1330 FOR I=1 TO N
1340 F5=F5+Z[I]^2*(F[I]*M[I]-Q*Q[I])*(V[I]-Q*U[I])/(M[I])^2
1350 NEXT I
1360 FOR I=1 TO N
1370 F4=-Q*Z[I]/S1+(2*F5/S1-X1)*(-T[I]/D9)
1380 F7=Z[I]/S1+(2*F5/S1-X1)*(-E[I]/D9)
1390 Z4=Z4+F4^2/P[I]
1400 H1=H1+T[I]*E[I]*F[I]/M[I]
1410 H2=H2+F4*F7*F[I]/M[I]
1420 S2=S2+F7^2/Q[I]
1430 NEXT I
1440 A=(N#2)*S/(N-2+(N=2))
1450 F1=SQR(Z4+S2+2*H2)
1460 B3=(SQR(A8+A9+2*H1))/D9
1470 B5=B3*SQRA
1480 S1=F1*SQRA
1490 FIND #I[2,1],I[2,2]
1500 GOTO (I3#0)*(1+Y5) OF 1690,1650
1510 DISP "CALCULATING CHI-SQUARE SOLUTION"
1520 IF S#0 AND ((N>20)*(A<2.9)+(N>7)*(A<3.9)+(A<16)) THEN 1550
1530 F= NOT S
1540 GOTO 1570
1550 F=((S/2)^((N-2)/2))*(EXP(-S/2))/FNT(N+2)*(1+FNRS)
1560 F=(1-F)*(F#0)
1570 I0=1+(F<0.01)+(F<0.05)+(F<0.3)
1580 WRITE (2,1800)

```

```

1590 PRINT "***MODEL 1 SOLUTION -- ASSUMES ALL SCATTER DUE TO ANALYTICAL ERROR**
1600 I5=P2=1.96*B3
1610 I4=P1=1.96*F1
1620 Q2=Q
1630 A6=A7
1640 GOTO 1770
1650 PRINT "*****MODEL 3 SOLUTION --ASSUMES SCATTER DUE TO ANALYTICAL ERROR"
1660 PRINT "*****PLUS NORMAL VARIATION IN INITIAL "F5"*****"LIN1
1670 PRINT LIN1"CALC. VARIATION IN INITIAL "F5"="2*S8" (2-SIGMA)"LIN1
1680 GOTO 1710
1690 PRINT "*****MODEL 2 SOLUTION -- ASSUMES EQUALLY WEIGHTED POINTS AND"
1700 PRINT "*****NORMAL DISTRIBUTION OF RESIDUALS"
1710 RESTORE 2370
1720 FOR I=1 TO N-2
1730 READ C0
1740 NEXT I
1750 P2=C0*B5
1760 P1=C0*S1
1770 PRINT LIN2,SPA33"SLOPE"SPA19"INTERCEPT"
1780 PRINT SPA33"-----"SPA19"-----"LIN2
1790 PRINT "BEST-FIT:"TAB31,Q;TAB56;A7
1800 FORMAT 90"-",/
1810 WRITE (2,1800)
1820 IF I3 THEN 1840
1830 PRINT "ERROR (1-SIGMA, A PRIORI) "TAB31,B3;TAB56;F1,LIN1
1840 PRINT "ERROR (1-SIGMA, INCL. SCATTER) "TAB31,B5;TAB56;S1
1850 PRINT LIN1"ERROR (95% CONF. LIMITS)"TAB31,P2;TAB56;P1,LIN1
1860 PRINT LIN1,"COORDINATES OF CENTROID: X="X1;"Y="Y1,LIN1
1870 IF I3 THEN 1940
1880 FORMAT "SUMS =",F7.1," -- FOLLOWS CHI-SQUARE DISTRIBUTION ABOUT",F3.0,/
1890 WRITE (2,1880)S,N-2
1900 FIND #I[2,1],I[2,2]
1910 PRINT "THE OBSERVED SCATTER WILL BE CAUSED BY THE ASSIGNED ANALYTICAL"
1920 FORMAT 15X,"ERRORS ALONE",F7.2,"% OF THE TIME"
1930 WRITE (2,1920)100*F
1940 WRITE (2,1800)
1950 I3=1+I3
1960 IF I3=2 OR I0 <= 1 THEN 2000
1970 IF NOT Y5 THEN 410
1980 S8=S1
1990 GOTO 410
2000 D=2+Y5
2010 IF I0>1 AND I0<4 THEN 2040
2020 I0=(I0=1)+D*(I0=4)
2030 GOTO 2070
2040 DISP "MODEL 1 OR MODEL"D"SOLUTION";
2050 INPUT I0
2060 IF I0#1 AND I0#D THEN 2040
2070 Q=Q2*(I0=1)+Q*(I0=D OR I0=0)
2080 A7=A6*(I0=1)+A7*(I0=D OR I0=0)
2090 P2=P2*(I0=D OR I0=0)+I5*(I0=1)
2100 P1=P1*(I0=D OR I0=0)+I4*(I0=1)
2110 LINE 5,8
2120 DEF FN(X,Z1)=Q*Z1+A7

```

```

2130 DEF FNY(Z1)=(Z1-A7)/Q
2140 GOTO 1+(FNXA1>B2)+2*(FNXA1<A2) OF 2150,2170,2190
2150 PLOT A1,FNXA1
2160 GOTO 2200
2170 PLOT FNYB2,B2
2180 GOTO 2200
2190 PLOT FNYA2,A2
2200 GOTO 1+(FNXB1>B2)+2*(A2>FNXB1) OF 2210,2230,2250
2210 PLOT B1-0.01*(B1-A1),FNX(B1-0.01*(B1-A1)),-1
2220 GOTO 2260
2230 PLOT FNY(0.99*(B2-A2)+A2),0.99*(B2-A2)+A2,-1
2240 GOTO 2260
2250 PLOT FNY(A2+0.01*(B2-A2)),A2+0.01*(B2-A2),-1
2260 LINE
2270 MAT R=ZER
2280 MAT U=ZER
2290 MAT S=ZER
2300 LINK #I[2,1],I[2,2]
2310 I3=1+I3
2320 PRINT LIN1"***DATA CANNOT BE YORKFIT WITH MODEL"I3+(I3#1)*Y5"ASSUMPTIONS"
2330 PRINT "****"LIN2
2340 IF I3=1 THEN 400
2350 I0=1
2360 GOTO 2070
2370 DATA 12.71,4.3,3.18,2.78,2.57,2.45,2.36,2.31,2.26,2.23,2.2,2.18,2.16
2380 DATA 2.14,2.13,2.12,2.11,2.1,2.09,2.09,2.08,2.07,2.07,2.06,2.06,2.06,2.05
2390 DATA 2.05,2.05,2.04,2.04,2.04,2.04,2.03,2.03,2.03,2.03,2.02,2.02,2.02
2400 DATA 2.02,2.02,2.02,2.02,2.02,2.02,2.01,2.01
2410 END

```

APPENDIX V -- LIST OF PART 4 OF PROGRAM

```

10 REM INSTRUCTIONS PROGRAM, GENERAL PLOTTER
20 FIND #I[1,1],I[1,2]
30 PRINT "INSTRUCTIONS FOR K.R. LUDWIG HP-9831/9872 GENERAL PLOTTING PROGRAM"
40 PRINT TAB10"U.S. Geological Survey Open-File Rept. #79-1641"LIN3
50 PRINT "INTRODUCTION"LIN1"-----"LIN2
60 PRINT TAB6"This program is written for an HP-9831 computer with 12K words"
70 PRINT "memory, and an HP-9872 plotter. Data may be entered either from the"
80 PRINT "keyboard, or from data stored in a split-precision, 50X35 array as"
90 PRINT "created by the K.R.L. data-file program (U.S.G.S. Open-File Rept. #"
100 PRINT "79-1434). X- and Y-axis labels may either be taken from the parameter
110 PRINT "names defined in the data-file, or keyboard-input. Data-points may"
120 PRINT "be plotted using a variety of symbols, including open/solid plygons,
130 PRINT "error-boxes, error-crosses, or error-ellipses. Errors and error-"
140 PRINT "correlations may be input either from a data-file, or from the key-"
150 PRINT "board, even if the data are from a data-file. Plot-size and relative
160 PRINT "dimensions may be adjusted by the user over a wide range without"
170 PRINT "distortion of symbols. Once plotted, a least-squares regression line
180 PRINT "may be calculated and plotted, using the standard or one of two"
190 PRINT "modified York-type algorithms. Isochron ages and errors for most"
200 PRINT "isotope-ratio plots (including 207-Pb/204-Pb vs. 206-Pb/204-Pb) may"
210 PRINT "be calculated from the regression-line data."LIN3
220 PRINT "  DISPLAY"TAB40,"RESPONSE"
230 PRINT "  -----"TAB40,"-----"LIN2
240 PRINT "DATA FROM TRACK, FILE#?"TAB31"If you wish to plot data from a data-"
250 PRINT TAB30"file, enter the track and file numbers"
260 PRINT TAB30"of that data-file. If data are to be"
270 PRINT TAB30"entered only from the keyboard, enter"
280 PRINT TAB30"a space."LIN2
290 PRINT "SIZE: 1=STD, (2)=SMALL, 3=INSTR.?      Enter 1 for a plot scaled"
300 PRINT TAB30"to fit on a horizontal, 8 x 10.5 in."
310 PRINT TAB30"sheet of paper; enter 2 for a smaller"
320 PRINT TAB30"plot expressly scaled for reproduction as"
330 PRINT TAB30"a 35mm transparency or journal-figure"
340 PRINT TAB30"(or -2 for a plot with arbitrary size"
350 PRINT TAB30"but relatively larger labels and symbols);"
360 PRINT TAB30"enter 3 to obtain a printout of these"
370 PRINT TAB30"instructions. Any other number permits"
380 PRINT TAB30"the user to adjust the size of the plot,"
390 PRINT TAB30"using the P1-P2 controls of the plotter."LIN2
400 PRINT "X-AXIS (NAME OR PARAMETER #? If data are to be taken mainly from"
410 PRINT TAB30"a data-file, enter the parameter number"
420 PRINT TAB30"of the X-axis data. Otherwise,"
430 PRINT TAB30"enter a label from the keyboard. If you"
440 PRINT TAB30"wish all numerals of the label to be"
450 PRINT TAB30"superscripts, enter ^ as the first"
460 PRINT TAB30"character of the label (e.g., ^87Rb/86Sr)."
470 PRINT TAB30"If you wish the numerals to be subscripts,"
480 PRINT TAB30"enter an asterisk (*) as the first"
490 PRINT TAB30"character (e.g., *Al2O3)."LIN2
500 PRINT LIN2,"Y-AXIS (NAME OR PARAMETER #)?"TAB30"Same input as for X-axis."

```

```

510 PRINT LIN2"X AND Y LIMITS?"TAB30"Enter 4 values: X-MIN, X-MAX, Y-MIN,"
520 PRINT TAB30"Y-MAX, to define the limits of the plot."LIN3
530 PRINT "A list summarizing the symbol options now appears as printout."LIN2
532 PRINT "An input of '@' will restart the plot set-up part of the program."Lin
540 PRINT "An input of 'p' or 'P' will permit any regular polygon of any orien
550 PRINT "tation to be used as a plotting-symbol. The later 'ROTATION' reques
560 PRINT "asks for the angle in degrees (3 o'clock=0 degrees, 12 o'clock=90"
570 PRINT "degrees) of a vertex of the polygon, 'SIZE' request asks for relative
580 PRINT "size of the polygon. Values of 2 to 3 are most useful. Size is"
590 PRINT "proportional to the number entered. P gives solid, p open symbols.
600 PRINT LIN2"An input of 'e' or 'E' will give open or solid error-rectangles"
610 PRINT "as plotting-symbols."LIN2
620 PRINT "An input of 'c' or 'C' will give open or solid error-ellipses as"
630 PRINT "plotting-symbols. You will need to know the correlation-coefficient
640 PRINT "of the X and Y errors (0 if uncorrelated)."LIN2
650 PRINT "An input of '.' will give error-crosses as plotting symbols."LIN2
660 PRINT "An input of 'L' or 'l' will permit you to letter any statement"
670 PRINT "anywhere on the plot. The subsequent 'SIZE (.5-2), (ANGLE), (SLANT)
680 PRINT "display requests a number defining the relative size of the letters
690 PRINT "(0.5 to 2 is the most useful range); the angle of the line along which
700 PRINT "the phrase is to be lettered, in degrees; and the slant of the"
710 PRINT "letters (also in degrees). Note that input of the latter two"
720 PRINT "parameters is optional -- only the letter-size parameter, or the"
730 PRINT "letter-size plus line-angle parameters need be entered. The next"
740 PRINT "display will be, 'POSITION PEN. (LIVE KEYBOARD)'. At this point,"
750 PRINT "you may position the plotter-pen by using the four 'arrow' keys at"
760 PRINT "the top of the keyboard. DO NOT PFESS A CHARACTER KEY, or the char-
770 PRINT "acter will be immediately printed on the plot. Press the STOP key"
780 PRINT "when the pen is positioned correctly. Then type in the statement to
790 PRINT "be lettered, and press the 'EXECUTE' key. To have the phrase letter
800 PRINT "on successively lower lines, include the two-character symbol '//'"
810 PRINT "wherever you wish the lettering of the phrase to be repositioned one
820 PRINT "line lower than before, starting just below the first character of"
830 PRINT "the phrase. If the three-character symbol '+/-' is included in the"
840 PRINT "phrase, it will appear as a '+' over a '-' symbol, as is usually"
850 PRINT "grafted, and will occupy one character space."LIN2
860 PRINT "An input of 'A' or 'a' will result in the calculation of an isochron
870 PRINT "age from a just-calculated Yorkfit slope. Enter the appropriate"
880 PRINT "decay constant (in units of reciprocal years) when requested. How-"
890 PRINT "ever, if the plot is 207-Pb/204-Pb vs 206-Pb/204-Pb, a 207/206 age"
900 PRINT "will be calculated using the IUGS decay constants (207 and 204 must"
910 PRINT "appear in the Y-label and 206 and 204 in the X-label)."LIN2
920 PRINT "An input of '^' will result in a line connecting the data-points,"
930 PRINT "in the order that they were entered."LIN2
940 PRINT "Any other UPPER CASE keyboard symbol may also be used as a plotting-
950 PRINT "symbol (such as X, *, +, O). Lower-case symbols and other asymmet-"
960 PRINT "rical characters will not be plotted accurately, however."LIN3
970 PRINT "DISPLAY"TAB40"RESPONSE"LIN1"-----"TAB40"-----"LIN2
980 PRINT "X, (ERR,) Y (,ERR) (,CORRELATION)? If keyboard input, the data for"
990 PRINT TAB30"each point may be the"
1000 PRINT TAB30"X and Y values, (2 values separated by"
1010 PRINT TAB30"a comma), as the X and Y values together"
1020 PRINT TAB30"with their errors (X, X-error, Y,Y-error)"
1030 PRINT TAB30"or as the X and Y values together with"

```



```

1040 PRINT TAB30"their errors and with error-correlations"
1050 PRINT TAB30"(X, X-error, Y, Y-error, error-correl.)."
1060 PRINT TAB30"If error or error-correlation infor-"
1070 PRINT TAB30"mation that was not entered is needed"
1080 PRINT TAB30"for later plotting or Yorkfit purposes,"
1090 PRINT TAB30"it will be requested later."LIN2
1100 PRINT TAB30"If the data is to be input from a data-"
1110 PRINT TAB30"file, input a asterisk followed by the"
1120 PRINT TAB30"number(s) of the sets to be plotted."
1130 PRINT TAB30"An input of *8 will input set 8, whereas"
1140 PRINT TAB30"an input of *9,42 would input sets 9"
1150 PRINT TAB30"through 42, inclusive, from the data-file"
1160 PRINT TAB30"in memory. If you wish to change"
1170 PRINT TAB30"the data-file in memory, input 'DF' or"
1180 PRINT TAB30"'df' and the display will request the file"
1190 PRINT TAB30"number of the new data-file."LIN2
1200 PRINT TAB30"When all data for the particular plot-"
1210 PRINT TAB30"ting-symbol has been entered, enter"
1220 PRINT TAB30"a space."LIN3
1230 PRINT "?EDIT ENTRY#? (0 IF OK)?"TAB30"If your data was input correctly,"
1240 PRINT TAB30"enter 0. If corrections are necessary,"
1250 PRINT TAB30"enter the number of the set to be edited."LIN3
1260 PRINT "DATA TO BE RETAINED (1=YES)? If you wish the just-plotted data to"
1270 PRINT TAB30"be combined with subsequently input data"
1280 PRINT TAB30"(perhaps with different plotting symbols)"
1290 PRINT TAB30"for a later Yorkfit, input 1. Other-"
1300 PRINT TAB30"wise, input any other number."LIN3
1310 PRINT "X-ERRORS, Y-ERRORS (LAST BLOCK) This display appears if required"
1320 PRINT TAB30"error-information has not been entered."
1330 PRINT TAB30"For keyboard error-input, enter the"
1340 PRINT TAB30"X- and Y-errors, separated by a comma."
1350 PRINT TAB30"If the errors are stored in the data-"
1360 PRINT TAB30"file, however. input an asterisk,"
1370 PRINT TAB30"followed by the parameter numbers of the"
1380 PRINT TAB30"X- and Y-errors (e.g., *10,16)."LIN2
1390 PRINT "NOTE: ALL ERRORS MUST BE INPUT IN PERCENT AND, FOR YORKFIT INPUT,"
1400 PRINT TAB8"AT THE 2-SIGMA LEVEL."LIN3
1410 PRINT "ERROR-CORRELATION (LAST BLOCK)? Similar response to above."LIN2
1420 PRINT "NAME FOR THIS DATA SET?"TAB30"Enter a name for the Yorkfit printout"
1430 PRINT LIN3,TAB15"REMARKS ON YORKFIT SOLUTIONS"LIN2
1440 PRINT " The Yorkfit routine may be done in three ways. The first"
1450 PRINT "solution (York's original algorithm) weights the points according to"
1460 PRINT "their analytical errors, and calculates the best-fit line and errors"
1470 PRINT "assuming that the only cause of scatter from a straight line is"
1480 PRINT "analytical error. However, this may not be realistic if the "
1490 PRINT "calculated probability (from the comparison of the sums of the"
1500 PRINT "squares of the residuals (SUMS) with the degrees of freedom) is low"
1510 PRINT "This probability corresponds to how often one may expect the data,"
1520 PRINT "with their associated analytical errors, to give the observed"
1530 PRINT "amount of scatter. If this probability is less than 30%, the"
1540 PRINT "Yorkfit will be redone by assuming either equally-weighted"
1550 PRINT "points with zero error-correlations (model 2), or by assuming that"
1560 PRINT "the cause of the excess scatter is due to an additional variance in"
1570 PRINT "the Y-parameter (model 3). The model 2 solution assumes only that"

```

```
1580 PRINT "the cause for scatter has a normal distribution. Model 3 corres-"
1590 PRINT "ponds to the variable initial-ratio problem of Rb-Sr isochrons. The
1600 PRINT "model 3 solution will be calculated whenever the Y-axis name in-"
1610 PRINT "includes the characters 87Sr/86Sr, 143Nd/144Nd, 206Pb/204Pb, or"
1620 PRINT "207Pb/204Pb when the X-axis is not 206Pb/204Pb. The solution will"
1630 PRINT "also calculate the most probable value of the standard-deviation of
1640 PRINT "the initial Y. If the model 1 probability is greater than 30%, only
1650 PRINT "the model 1 solution will be solved. If the probability is between
1660 PRINT "1% and 30%, the user can choose which solution to use. If the"
1670 PRINT "probability is less than 1%, only the model 2 or 3 solution can be"
1680 PRINT "used (for plotting of the best-fit line and for later isochron-age"
1690 PRINT "solutions). If a model 2 solution is obtained when a model 3 so-"
1700 PRINT "lution is desired (or vice-versa), the correct model-solution can"
1710 PRINT "be obtained by executing 'SFLAG1' before running the program."LIN4
1720 LOAD #I[1,1],I[1,2]
1730 END
```